Scatter and Glow

By Dror Bar-Natan . . . . . , December 2008.

Project goals

- Verify R3 and R2.
- Recover the Alexander polynomial of all knots.
- Recover the multi-variable Alexander polynomial of all links.
- The scatter and glow of an arbitrary exponential.
- Find an explicit BCH formula.
- Solve R4 for F at the scatter level.
- Verify the pentagon.
- Solve for F at the glow level.
- Check the Hexagons.
- Solve the \theta-R-F equation.
- Verify the Hexagons.
- Recover the Lieberum formulas.

Conventions

Ar[i, j] is an arrow going from i to j.
Y[i,j,k] := [Ar[i,k], Ar[j,k]] = Ar[i,k]Ar[j,k] - Ar[j,k]Ar[i,k] =: Ad[Ar[i,k]][Ar[j,k]] = -[Ar[i,j], Ar[j,k]]
x[l]Y[i,j,k] := [Ar[l,k], Y[i,j,k]]
IHX: x[i]Y[j,k,l]+x[j]Y[k,i,l]+x[k]Y[i,j,l] = 0

Program

In[1]:=
```
S[sigma[i_, j_]] := S[{
     Ar[0, j] → Ar[0, j] + Y[0, i, j, -(Exp[-x[i]] - 1) / x[i]],
     Ar[0, i] → Ar[0, i] + Y[0, i, j, (Exp[-x[i]] - 1) / x[i]],
     Ar[j, 0] → Ar[j, 0] + Y[i, j, 0, (Exp[x[i]] - 1) / x[i]]
    }];

ReducePrimitives[prims_] := Module[{l, h0, h1}, prims
     //. {
      Y[__, 0] → 0,
      Y[i_, i_, __] → 0,
      Y[i_, j_, k_, h_] /; i > j :→ Y[j, i, k, Expand[-h]],
      c_ * Y[i_, j_, k_, h_] :→ Y[i, j, k, Expand[c * h]],
      Y[i_, j_, k_, h1_] + Y[i_, j_, k_, h2_] :→ Y[i, j, k, h1 + h2],
      Y[i_, j_, k_, h_] /; ! FreeQ[h, x[l_] /; l < i] :→ (
        l = Min[Cases[{h}, x[l_] :→ l, Infinity]];
        h0 = Limit[h, x[l] → 0];
        h1 = Expand[(h - h0) / x[l]];
        Y[i, j, k, h0]
         - Y[j, l, k, Expand[h1 x[i]]] - Y[l, i, k, Expand[h1 * x[j]]]
       )
     }
    /. Y[i_, j_, k_, h_] :→ Y[i, j, k, Expand[h]]
  ];

Scatter[S[s_List]][prims_] := ReducePrimitives[prims
      /. {
       Ar[i_, j_] :→ Distribute[Ar[Ar[i, 0] /. s, Ar[0, j] /. s]],
       Y[i_, j_, k_, h_] :→ Distribute[Y[
          Ar[i, 0] /. s, Ar[j, 0] /. s, Ar[0, k] /. s, h
         ]]
      }
     /. {Ar[i_, 0] :→ i, Ar[0, j_] :→ j}
    /. {
     Ar[Y[i_, j_, 0, h_], k_] :→ Y[i, j, k, h],
     Y[_Y, _Y, __] :→ 0,
     Y[i_Integer, Y[j_, k_, 0, h_], l_, h1_] :→ Y[j, k, l, x[i] * h * h1],
     Y[Y[j_, k_, 0, h_], i_Integer, l_, h1_] :→ Y[j, k, l, -x[i] * h * h1]
    }
   /. {
    Ar[i_, Y[0, j_, k_, h_]] :→ Y[i, j, k, h],
    Ar[i_, Y[j_, 0, k_, h_]] :→ Y[j, i, k, h],
    Y[i_, j_, Y[0, k_, l_, h_], h1_] :→ Y[i, j, l, -x[k] * h * h1],
    Y[i_, j_, Y[k_, 0, l_, h_], h1_] :→ Y[i, j, l, x[k] * h * h1]
   }
 ]
```

Keep S as only operator.

Testing

- **The braid group on two strands is commutative :**

In[4]:=
```
Scatter[S[sigma[1, 2]]][Ar[1, 2]] // ReducePrimitives
```

Out[4]=
```
Ar[1, 2]
```

- **Locallity in Scale (only for overcrossings)**

In[5]:=
```
(Ar[1, 2] // Scatter[S[sigma[3, 1]]]) // Scatter[S[sigma[3, 2]]]
```

Out[5]=
```
Ar[1, 2] + Y[1, 3, 2, 0]
```

In[6]:=
```
(Ar[2, 1] // Scatter[S[sigma[3, 1]]]) // Scatter[S[sigma[3, 2]]]
```

Out[6]=
```
Ar[2, 1]
```

- **Overcrossings Commute**

In[7]:=
```
oc1 = {Ar[1, 4], Ar[2, 4], Ar[3, 4], Ar[4, 1], Ar[4, 2], Ar[4, 3], Ar[1, 2], Y[1, 2, 3, 1],
    Y[2, 3, 1, 1], Y[3, 1, 2, 1]} // Scatter[S[sigma[1, 2]]] // Scatter[S[sigma[1, 3]]]
```

Out[7]=
$$\left\{\text{Ar}[1,4],\ \text{Ar}[2,4]+\text{Y}\left[1,2,4,-\frac{1}{x[1]}+\frac{e^{x[1]}}{x[1]}\right],\ \text{Ar}[3,4]+\text{Y}\left[1,3,4,-\frac{1}{x[1]}+\frac{e^{x[1]}}{x[1]}\right],\right.$$
$$\text{Ar}[4,1]+\text{Y}\left[1,4,2,\frac{1}{x[1]}-\frac{e^{-x[1]}}{x[1]}\right]+\text{Y}\left[1,4,3,\frac{1}{x[1]}-\frac{e^{-x[1]}}{x[1]}\right],$$
$$\text{Ar}[4,2]+\text{Y}\left[1,4,2,-\frac{1}{x[1]}+\frac{e^{-x[1]}}{x[1]}\right],\ \text{Ar}[4,3]+\text{Y}\left[1,4,3,-\frac{1}{x[1]}+\frac{e^{-x[1]}}{x[1]}\right],\ \text{Ar}[1,2],$$
$$\text{Y}[1,2,3,1],\ \text{Y}\left[1,2,1,\frac{x[3]}{x[1]}-\frac{e^{x[1]}\,x[3]}{x[1]}\right]+\text{Y}\left[1,2,2,\frac{x[3]}{x[1]}-\frac{e^{x[1]}\,x[3]}{x[1]}\right]+$$
$$\text{Y}\left[1,2,3,\frac{x[3]}{x[1]}-\frac{e^{x[1]}\,x[3]}{x[1]}\right]+\text{Y}\left[1,3,1,-\frac{x[2]}{x[1]}+\frac{e^{x[1]}\,x[2]}{x[1]}\right]+$$
$$\left.\text{Y}\left[1,3,2,-\frac{x[2]}{x[1]}+\frac{e^{x[1]}\,x[2]}{x[1]}\right]+\text{Y}\left[1,3,3,-\frac{x[2]}{x[1]}+\frac{e^{x[1]}\,x[2]}{x[1]}\right]+\text{Y}[2,3,1,1],\ \text{Y}[1,3,2,-1]\right\}$$

In[8]:=
```
oc2 = {Ar[1, 4], Ar[2, 4], Ar[3, 4], Ar[4, 1], Ar[4, 2], Ar[4, 3], Ar[1, 2], Y[1, 2, 3, 1],
    Y[2, 3, 1, 1], Y[3, 1, 2, 1]} // Scatter[S[sigma[1, 3]]] // Scatter[S[sigma[1, 2]]];
Thread[
 oc1 ==
  oc2]
```

Out[9]=
```
True
```

In[10]:=
```
t1 =
 {Ar[1, 4], Ar[2, 4], Ar[3, 4], Ar[4, 1], Ar[4, 2], Ar[4, 3]} // Scatter[S[sigma[1, 2]]] //
  Scatter[S[sigma[1, 3]]] // Scatter[S[sigma[2, 3]]]
```

Out[10]=
$$\left\{\text{Ar}[1,4],\ \text{Ar}[2,4]+\text{Y}\left[1,2,4,-\frac{1}{x[1]}+\frac{e^{x[1]}}{x[1]}\right],\right.$$
$$\text{Ar}[3,4]+\text{Y}\left[1,2,4,-\frac{x[3]}{x[1]\,x[2]}+\frac{e^{x[1]}\,x[3]}{x[1]\,x[2]}+\frac{e^{x[2]}\,x[3]}{x[1]\,x[2]}-\frac{e^{x[1]+x[2]}\,x[3]}{x[1]\,x[2]}\right]+$$
$$\text{Y}\left[1,3,4,-\frac{e^{x[2]}}{x[1]}+\frac{e^{x[1]+x[2]}}{x[1]}\right]+\text{Y}\left[2,3,4,-\frac{1}{x[2]}+\frac{e^{x[2]}}{x[2]}\right],$$
$$\text{Ar}[4,1]+\text{Y}\left[1,4,2,\frac{1}{x[1]}-\frac{e^{-x[1]}}{x[1]}\right]+\text{Y}\left[1,4,3,\frac{1}{x[1]}-\frac{e^{-x[1]}}{x[1]}\right],$$
$$\text{Ar}[4,2]+\text{Y}\left[1,4,2,-\frac{1}{x[1]}+\frac{e^{-x[1]}}{x[1]}\right]+\text{Y}\left[1,4,3,-\frac{1}{x[1]}+\frac{e^{-x[1]}}{x[1]}-\frac{e^{-x[1]-x[2]}}{x[1]}+\frac{e^{-x[2]}}{x[1]}\right]+$$
$$\left.\text{Y}\left[2,4,3,\frac{1}{x[2]}-\frac{e^{-x[2]}}{x[2]}\right],\ \text{Ar}[4,3]+\text{Y}\left[1,4,3,\frac{e^{-x[1]-x[2]}}{x[1]}-\frac{e^{-x[2]}}{x[1]}\right]+\text{Y}\left[2,4,3,-\frac{1}{x[2]}+\frac{e^{-x[2]}}{x[2]}\right]\right\}$$

In[11]:=
```
t2 =
 {Ar[1, 4], Ar[2, 4], Ar[3, 4], Ar[4, 1], Ar[4, 2], Ar[4, 3]} // Scatter[S[sigma[2, 3]]] //
  Scatter[S[sigma[1, 3]]] // Scatter[S[sigma[1, 2]]]
```

Out[11]=
$$\left\{\text{Ar}[1,4],\ \text{Ar}[2,4]+\text{Y}\left[1,2,4,-\frac{1}{x[1]}+\frac{e^{x[1]}}{x[1]}\right],\right.$$
$$\text{Ar}[3,4]+\text{Y}\left[1,2,4,-\frac{x[3]}{x[1]\,x[2]}+\frac{e^{x[1]}\,x[3]}{x[1]\,x[2]}+\frac{e^{x[2]}\,x[3]}{x[1]\,x[2]}-\frac{e^{x[1]+x[2]}\,x[3]}{x[1]\,x[2]}\right]+$$
$$\text{Y}\left[1,3,4,-\frac{e^{x[2]}}{x[1]}+\frac{e^{x[1]+x[2]}}{x[1]}\right]+\text{Y}\left[2,3,4,-\frac{1}{x[2]}+\frac{e^{x[2]}}{x[2]}\right],$$
$$\text{Ar}[4,1]+\text{Y}\left[1,4,2,\frac{1}{x[1]}-\frac{e^{-x[1]}}{x[1]}\right]+\text{Y}\left[1,4,3,\frac{1}{x[1]}-\frac{e^{-x[1]}}{x[1]}\right],$$
$$\text{Ar}[4,2]+\text{Y}[1,2,3,0]+\text{Y}\left[1,4,2,-\frac{1}{x[1]}+\frac{e^{-x[1]}}{x[1]}\right]+$$
$$\text{Y}\left[1,4,3,-\frac{1}{x[1]}+\frac{e^{-x[1]}}{x[1]}-\frac{e^{-x[1]-x[2]}}{x[1]}+\frac{e^{-x[2]}}{x[1]}\right]+\text{Y}\left[2,4,3,\frac{1}{x[2]}-\frac{e^{-x[2]}}{x[2]}\right],$$
$$\left.\text{Ar}[4,3]+\text{Y}[1,2,3,0]+\text{Y}\left[1,4,3,\frac{e^{-x[1]-x[2]}}{x[1]}-\frac{e^{-x[2]}}{x[1]}\right]+\text{Y}\left[2,4,3,-\frac{1}{x[2]}+\frac{e^{-x[2]}}{x[2]}\right]\right\}$$

In[12]:=
```
ReducePrimitives[t1 - t2]
```

Out[12]=
```
{0, 0, 0, 0, 0, 0}
```

Mathematica Experiments

"Testing" File, to be renamed "Experiments"